



This work is licensed under a [Creative Commons Attribution-NonCommercial-NoDerivs 2.5 License](http://creativecommons.org/licenses/by-nc-nd/2.5/).

Fundo Flutuante com transição de imagem Flash AS2

Este tutorial funciona como neste exemplo:

http://www.flashopen.net/tutorials/bg_float_multi/

Para o efeito, a imagem deve exceder as medidas do browser...

No exemplo as imagens têm 2800x2100px


Cuidado! O Flash ignora excessos de pixels acima dos 2880px ...

Não quer dizer q não tenha solução... mas isso ficará para o próximo tutorial ;)

Para que tudo funcione direitinho, todas as imagens devem ter as mesmas dimensões

1. As imagens serão carregadas dinamicamente de um mapa chamado 'assets'
2. As imagens estão definidas num ficheiro xml que também é carregado dinamicamente

Se usa apenas a primeira frame da timeline e só para acomodar 3 layers do nosso AS:



Portanto não temos nada no palco ou na Library.

A cor do palco deve ser preto, mas isso fica também ao critério de cada um.

Vamos lá então:

Começemos pelo layer do **preloader** (que é um tutorial por si).

Leia os comentários no próprio AS marcados por //

```
/*  
    @author: docFlash  
    @date: March 2009  
*/  
Stage.showMenu = false;  
//  
// -----> 1. declarar e setar variaveis
```

```

//
// - geral
// - pode alterar o scaleMode e altere os valores para as dimensoes do
seu Stage
//
var swfLargura:Number = 800;
var swfAltura:Number = 600;
//
// - variavel para o loader
// - nao alterar
//
var loader_counter:Number = 0;
var loader_animator:Number;
var loader_format:TextFormat = new TextFormat();
loader_format.font = "Arial";
loader_format.align = "center";
loader_format.color = 0xFFFFFFFF;
loader_format.size = 9;
//
// --- Iniciar --> ver 2
//
//
// -----> depth 5
//
// -----> 3. Desenhar o preloader
//
function draw_loader(target, nb, ba, lb, ab, dc, cb, ap, am, va) {
    //
    // 3a. criar movie principal
    var loader_rect:MovieClip =
createEmptyMovieClip("loader_rect",5);
    //
    // 3b. construir o preloader
    for (var i:Number = 0; i<nb; i++) {
        loader_rect.createEmptyMovieClip("loader_"+i,i+1);
        loader_rect["loader_"+i].beginFill(cb,am);
        loader_rect["loader_"+i].moveTo(-lb/2,-ab-dc);
        loader_rect["loader_"+i].lineTo(lb/2,-ab-dc);
        loader_rect["loader_"+i].lineTo(lb/2,-dc);
        loader_rect["loader_"+i].lineTo(-lb/2,-dc);
        loader_rect["loader_"+i].lineTo(-lb/2,-ab-dc);
        loader_rect["loader_"+i].endFill();
        //
        loader_rect["loader_"+i]._alpha = ap;
        loader_rect["loader_"+i]._rotation = i*(360/nb);
    }
    //
    loader_rect.createTextField("txt", loader_rect.getNextHighestDepth
(),-dc,-dc/2+4,2*dc,25);
    loader_rect.txt.text = "";
    loader_rect.txt.setTextFormat(loader_format);
    //
    // 3c. setar a posicao do preloader
    loader_rect._x = swfLargura/2-(lb+dc)/2;
    loader_rect._y = swfAltura/2-(ab+dc)/2;
    //
}

```

```

        // 3d. iniciar a animacao do preloader
        loader_animator =
setInterval(animate_loader,va,target,nb,ba,am,ap);
        // --- loop --- ver 4
    }
    //
    //
    // -----> 4. Animar o preloader
    //
function animate_loader(target, nb, ba, am, ap) {
    //
    i_index = 0;
    //
    // 4a. determinar quais as barras afectadas pelo alpha
    for (var i:Number = 0; i<ba; i++) {
        if (i+loader_counter<nb) {
            movieIndex = i+loader_counter;
        } else {
            movieIndex = i_index;
            i_index++;
        }
        //
        // setar o alpha as barras
        loader_rect["loader_"+movieIndex]._alpha = Math.round(i*((am-
ap)/(nb/2-1))+ap);
    }
    //
    // 4b. determinar a barra da frente
    if (loader_counter == nb-1) {
        loader_counter = 0;
    } else {
        loader_counter++;
    }
}
//
// -----> 5. Delete preloader
//
function delete_loader(nb) {
    loader_rect.txt.removeTextField();
    for (var i:Number = 0; i<nb; i++) {
        loader_rect["loader_"+i].removeMovieClip();
    }
    loader_rect.removeMovieClip();
    _root.mc.onEnterFrame = null;
    //trace("hum?");
}
//
//
// -----> 2. Iniciar criacao do
preloader
//
// draw_loader(nome do movieClip ou _root,
// numero de barras,
// barras afetadas pelo alpha,
// largura da barra,
// altura da barra,
// distancia ao centro,'

```

```

// cor da barra,
// alpha + pequeno,
// alpha maior,
// velocidade da animacao);
//
//
// [_root]
//
var n_barras:Number = 36;
draw_loader(_root,24,8,3,3,20,0xFFFFFFFF,20,100,30);
//
// ---- desenhar ---> ver 3
//
function loadingFilter(){
    valor = 1;
    crescendo = true;
    clean_preloader();
    _root.mc.onEnterFrame = function(){

        if(valor < 32 && crescendo == true){
            valor *= 2;
        }else if(valor > 0){
            crescendo = false;
            valor -= 2;
        }else{
            crescendo = true;
        }
        _root.make_allfilters(this,valor);
    };
}

```

O layer do **switch** pois aqui tudo se inicia.

Por falar em iniciar, repare que em primeiro lugar carregaremos a xml.

Depois teremos os nomes das imagens disponiveis,e aí iremos carregar uma por uma.

Comece no comentário INIT e tente compreender, para mais tarde poder fazer as modificações necessárias ou saber como e onde adicionar outros elementos do seu projecto.

```

//
// Props:
//   width=800;
//   height=600;
//   FrameRate=30fps;
//   Player=9
//   ActionScript=2.0
//
System.security.allowDomain("http://farm5.static.flickr.com");
//
// stage mode
Stage.scaleMode = "noscale";
//
// stage zeros
var zero_x:Number = -(Stage.width-800)/2;
var zero_y:Number = -(Stage.height-600)/2;
//
// stage listener
var stageobj:Object = new Object();
Stage.addListener(stageobj);

```

```

//
// -----Background
//
// movieclip loader
var mc_loader:MovieClipLoader = new MovieClipLoader();
var mc_object:Object = new Object();
mc_loader.addListener(mc_object);
//
// main mc
this.createEmptyMovieClip("mc",1);
mc.createEmptyMovieClip("mc",1);
mc.createEmptyMovieClip("mc1",2);
//
var presentClip:MovieClip = mc.mc;
var pastClip:MovieClip;
var countClip:Number = 0;
//
// paths
var image_url:String = "assets/";
var src:String = "";
//
// cargar intervalo
var intervalo:Number = 5000;//segundos
//
// first time
var first_time:Boolean = true;
//
// Load Background
function load_clip() {
    //
    // load clip
    mc_loader.loadClip(_root.src,presentClip);
    presentClip._alpha = 0;
    //
    // load complete -> initiate
    mc_object.onLoadInit = function(img) {
        //

        // Is the first time
        if (_root.first_time == true) {
            //
            clean_preloader();
            //
            // center main mc
            mc._x = zero_x-(img._width-Stage.width)/2;
            mc._y = zero_y-(img._height-Stage.height)/2;
            img._alpha = 100;
            //
            _root.first_time = false;
            Mouse.addListener(mouse_move);
        } else {
            image_alpha();
        }
        //
        setTimeout(load_morebgs,intervalo);
        //
        delete this.onLoadInit;
    }
}

```

```

        delete this.onLoadProgress;
        delete this.onLoadError;
    };
    //
    // error loading
    mc_object.onLoadError = function(img, code, status) {
        trace(img);
        trace(code);
        trace(status);
        //
        delete this.onLoadInit;
        delete this.onLoadProgress;
        delete this.onLoadError;
    };
    //
    // loading progress...
    mc_object.onLoadProgress = function(target:MovieClip,
bytesLoaded:Number, bytesTotal:Number) {
        loader_rect.txt.text =
Math.round(bytesLoaded*100/bytesTotal)+" %";
        loader_rect.txt.setTextFormat(loader_format);
    };
}
//
// random images to load
function load_morebgs() {
    countClip = countClip == bgxml.firstChild.childNodes.length-
1?0:countClip+1;
    trace(countClip);
    src =
_root.image_url+""+bgxml.firstChild.childNodes[countClip].attributes["s
rc"];
    //
    pastClip = presentClip;
    presentClip = presentClip == mc.mc?mc.mc1:mc.mc;
    presentClip.swapDepths(pastClip);
    //
    load_clip(presentClip);
}
//
// alpha transition
function image_alpha() {
    //
    _root.onEnterFrame = function() {
        if (presentClip._alpha>=100) {
            presentClip._alpha = 100;
            //
            delete _root.onEnterFrame;
            pastClip.unloadMovie();
        } else {
            presentClip._alpha += 5;
            pastClip._alpha -= 5;
        }
    };
}
//
//

```

```

// INIT - background
//
// XML
var bgxml:XML = new XML();
bgxml.ignoreWhite = true;
//
// load xml
bgxml.load("images_bg.xml");
//
bgxml.onLoad = function(success) {
    if (success) {
        mainNode = this.firstChild;
        mainLengt = this.firstChild.childNodes.length;
        //
        src =
image_url+" "+bgxml.firstChild.childNodes[countClip].attributes["src"];
        load_clip(presentClip);
    } else {
        getURL("javascript:alert('xml not found');");
    }
    delete this.onLoad;
};
//
// stage resize
stageobj.onResize = function() {
    zero_x = -(Stage.width-800)/2;
    zero_y = -(Stage.height-600)/2;

    mc.mc._alpha = 100;
    delete mc.onEnterFrame;
    move_photo(mc,100);
};
//
// clean preloader
function clean_preloader() {
    clearInterval(loader_animator);
    delete_loader(n_barras);
}

```

Por fim o layer **movebg** que é a nossa pequena máquina. Usando o método para o Mouse e calculando as bordas:

```

// variaveis para a velocidade
var speedX:Number;
var speedY:Number;
//
// variaveis para o destino
var destX:Number;
var destY:Number;
//
// Mouse foi acionado
//
var mouse_move:Object = new Object();
//
mouse_move.onMouseMove = function() {
    delete mc.onEnterFrame;
    updateAfterEvent();
}

```

```

        move_photo(mc,8);
};
//
// Mover imagem
//
function move_photo(mc, speed:Number) {
    //
    //calculos
    speedX = Math.round((xmouse(_xmouse)-
zero_x)/Stage.width*100)/100;
    speedY = Math.round((ymouse(_ymouse)-
zero_y)/Stage.height*100)/100;

    destX = Math.round(-((mc._width-Stage.width)*speedX));
    destY = Math.round(-((mc._height-Stage.height)*speedY));
    //
    // loop
    mc.onEnterFrame = function() {
        if (this._x != destX) {
            increaseX = Math.round((destX-
this._x+zero_x)*speed)/100;
            this._x += increaseX;
        }
        if (this._y != destY) {
            increaseY = Math.round((destY-
this._y+zero_y)*speed)/100;
            this._y += increaseY;
        }
        if (Math.abs(increaseX) < .5 && Math.abs(increaseY) < .5) {
            delete this.onEnterFrame;
        }
        updateAfterEvent();
    };
}
//
// mouse x
function xmouse(n):Number{
    if(n > zero_x+Stage.width){
        return zero_x+Stage.width;
    }else if(n < zero_x){
        return zero_x;
    }else{
        return n;
    }
}
//
// mouse y
function ymouse(n):Number{
    if(n > zero_y+Stage.height){
        return zero_y+Stage.height;
    }else if(n < zero_y){
        return zero_y;
    }else{
        return n;
    }
}
//

```

Pronto como o tempo é pouco para explicar linha por linha, considere este AS material de estudo.
Agora só falta deixar o ficheiro XML e como ele define as imagens para o Flash:

```
<images>
  <image src="P5181115.jpg" title="copo de vinho" description="" />
  <image src="PA061956_web.jpg" title="tomate e pepino"
description="" />
  <image src="PB152011.jpg" title="piano" description="" />
  <image src="PC130050.jpg" title="natal" description="" />
</images>
```

Neste caso não estamos usando o título nem a descrição, por isso considere como trabalho de casa!
Como dica, você sabe como gerar um campo de texto (no preloader) e como extraír informação da xml
(imagens). Bom trabalho e bom estudo... ah e já agora se tiver dificuldades pode sempre baixar os arquivos
aqui:

http://flashopen.nl/tutorials/bg_float_multi/flutTranAS2.zip

